


微型计算机图形学教程

微型计算机绘图

吴 忠



电子工业出版社

内 容 提 要

本书介绍 APPLE-II 微型计算机和 SR-6602 绘图仪的绘图指令及使用 BASIC 语言进行编程绘图的基本方法,为学习计算机绘图的同志提供一本入门性的科普读物。全书共分五章,包括:计算机绘图的基本方法;应用 BASIC 语言编写绘图程序;小型自动绘图系统的建立;机械图通用绘图子程序;绘图程序例题选等内容,约 20 学时即可学完。凡具有初中文化水平者,虽没接触过计算机,亦未学过 BASIC 语言,均可学会。

本书按国家教委职教司召开的全国中专教材规划会议精神编写,适应新的中专“机械制图”教学大纲中关于“微机绘图”的教学需要。本书初稿经北京中专制图教学研究会多次使用,效果显著。可供四年制中专校作为选修课教材,也可作为制图课中介绍微机绘图知识(4 学时)的补充教材。

与本书配套使用的有教学录像片一部,5.25 英寸软磁盘一片。

书末附有全部子程序清单供参考。

本书经北京中专制图教研会讨论并推荐出版。

微型计算机绘图

吴 忠

责任编辑:吴明卒

电子工业出版社出版(北京海淀区万寿路)

电子工业出版社发行 各地新华书店经售

中国科学院印刷厂印刷 印刷

*

开本:787×1092 毫米 1/16 印张:5.5 字数:134 千字

1988 年 7 月第一版 1988 年 7 月第一次印刷

印数:1—15000 册 定价:1.50 元

ISBN7-5053-0410-0/TP51

序

本书是学习计算机绘图入门性读物,以“初学者容易读懂”为宗旨,对象是未曾接触过计算机,未曾学过 BASIC 语言,具有初中文化水平的读者。通过本书的学习,读者可以基本掌握 BASIC 语言,熟悉计算机绘图的编程方法,为进一步学习微型计算机或从事计算机绘图工作,奠定基础。

本书摒弃系统的计算机理论和深奥的数学方法,从绘图编程的实践入手,介绍 BASIC 语言,做到语言与图形相结合,程序与作图步骤相结合,使初学者易于掌握。

全书共分五章:

第一章 计算机绘图的基本作图方法。以 APPLE-II 为样机,着重介绍基本绘图语句,主要是画直线的语句和画圆的语句。有了这两类基本语句,一般图形皆可作出。

第二章 以 APPLESOFT BASIC 为基础,介绍应用 BASIC 语言编写绘图程序。从绘图编程入手,学习 BASIC 语言及其应用,使语言与绘图程序紧密结合。每学一个语句,立即在绘图实践中应用。反之,每学一个绘图程序,也必然学到一个或几个 BASIC 语句。

第三章 小型自动绘图系统的建立。以 APPLE-II 与 SR-6602 自动绘图机联机为背景,介绍如何建立小型自动绘图系统,实现图形输出。

第四章 机械图通用绘图子程序。以 SR-6602 绘图机的绘图语句为基础,编制了 21 个基本图素的子程序,调用这些子程序,可绘制各类零件图、装配图、系统图。子程序存于磁盘内。

第五章 绘图程序例题选。介绍 APPLE-II 绘图程序和 SR-6602 绘图程序,其中有图案、平面图形、三视图、剖视图和零件图,可供初学编程者参考。

书中程序均上机通过,可确保无误。

本书编写过程中曾得到杜利生、邵二涟等同志的协助,在此一并致谢。

编著者

1987 年 6 月

目 录

第一章 计算机绘图的基本作图方法 .. 1	
§ 1-1 基本原理 .. 1	
§ 1-2 微机绘图系统简介 .. 2	
§ 1-3 绘图机的种类 .. 2	
§ 1-4 APPLE-II 基本绘图语句 .. 3	
第二章 应用 BASIC 语言编写 绘图程序 .. 9	
§ 2-1 基本 BASIC 语句简介 .. 9	
§ 2-2 循环(FOR / NEXT)语句 .. 9	
§ 2-3 输入(INPUT)语句 .. 12	
§ 2-4 读数(READ)语句和置数 (DATA)语句 .. 18	
§ 2-5 无条件转移(GOTO)语句 .. 19	
§ 2-6 条件转移(IF / THEN)语句 .. 19	
§ 2-7 转子(GOSUB)语句和返回 (RETURN)语句 .. 21	
§ 2-8 数组说明(DIM)语句 .. 24	
§ 2-9 其它基本 BASIC 语句 .. 26	
§ 2-10 扩展 BASIC 语句 .. 27	
第三章 小型自动绘图系统的建立 .. 29	
§ 3-1 SR-6602 绘图机性能介绍 .. 29	
§ 3-2 联机方式 .. 30	
§ 3-3 绘图指令 .. 30	
第四章 机械图通用绘图子程序 .. 38	
§ 4-1 画箭头(GOSUB 1400) .. 38	
§ 4-2 画图框、标题栏(GOSUB 1450) .. 38	
§ 4-3 字母 Φ(GOSUB 2000) .. 38	
§ 4-4 表面粗糙度符号(GOSUB 3000) .. 40	
§ 4-5 画倒角(GOSUB 4000) .. 40	
§ 4-6 画直线(GOSUB 5500) .. 40	
§ 4-7 画矩形或□型(GOSUB 5510) .. 40	
§ 4-8 画点划线(GOSUB 5530) .. 41	
§ 4-9 画一束水平线或竖直线 (GOSUB 5560) .. 41	
§ 4-10 画键槽(GOSUB 5600) .. 41	
§ 4-11 带圆角的矩形(GOSUB 5650) .. 41	
§ 4-12 绝对坐标画圆(GOSUB 6620) .. 42	
§ 4-13 绝对坐标画圆弧(GOSUB 6630) .. 42	
§ 4-14 相对坐标画圆和圆弧 (GOSUB 6640 和 GOSUB 6650) .. 42	
§ 4-15 定义线型(GOSUB 6700) .. 43	
§ 4-16 画圆环或圆的剖面(GOSUB 7000) 43	
§ 4-17 画键槽剖面(GOSUB 7500) .. 43	
§ 4-18 画矩形剖面 and 带倒角的矩形剖面 (GOSUB 9000) .. 43	
§ 4-19 画凸缘(GOSUB 9600) .. 44	
§ 4-20 绘图子程序的使用 .. 44	
第五章 绘图程序例题选 .. 46	
§ 5-1 APPLE-II 绘图程序 .. 46	
§ 5-2 SR-6602 绘图程序 .. 63	
附录 1: 上机操作 .. 73	
附录 2: 错误信息表 .. 75	
附录 3: 图形处理的矩阵变换公式 .. 77	
附录 4: 机械图通用绘图子程序清单 .. 78	

第一章 计算机绘图的基本作图方法

§ 1-1 基本原理

在过去的制图中,人们利用直尺或三角板来画直线,利用圆规来画圆或圆弧,利用曲线板或曲线尺来连接曲线,这些都是通过人的手来完成图形的。

在计算机绘图中,这些工具或仪器一概不用,而是通过程序对计算机发出各种指令,计算机又控制着绘图打印机(或 X-Y 绘图仪),使其按照人们的需要绘制出图形来。

一、图-数的转化

我们知道,任何图形都可以看作是满足某种条件的点的集合,而平面内的任何一点,均由两个坐标(X, Y)所确定。因此,图是可以用数来表示的。数字通过计算机都转化为电信号,这些电信号发出控制信息,准确无误地指挥着绘图打印机(或 X-Y 绘图仪)作机械运动,从而代替人工完成绘图。全部过程如图 1-1 所示。




图 1-1


二、画笔的基本动作

尽管绘图机的结构各不相同,但画笔的基本动作是一样的。对纸面来讲,画笔有四个基本运动方向+X、-X、+Y、-Y和四个与水平方向成 45° 的合成方向(+X, +Y)、(+X, -Y)、(-X, +Y)、(-X, -Y),如图 1-2 所示。

这八个方向的运动,每次成为一步,它与计算机的脉冲指令是一一对应的。这些脉冲指令驱动绘图机的步进电机一步步地转动,因此,称为步。

绘图机的画笔按脉冲指令画出的线条,是由一步步连接而组合起来的,如果步长小,就成为平滑的线段;如果步长大,就成为锯齿状的不平滑的折线。如图 1-3 所示,欲画线段 AB,画笔实际上是沿图中虚折线移动。由于步距很小,通常在 0.1 毫米以下,因此锯齿形走步是看不出来的,就宏观而论仍是平滑的图线。

画笔沿 X、Y 方向水平面移动的时候,也可沿与纸面垂直的 Z 方向上下运动,实现落笔画线和抬笔移动。上述八个方向的移动,叫做画笔的基本动作。



八个基本方向(S——步距)

图 1-2




图 1-3

§ 1-2 微机绘图系统简介

微机绘图系统的硬件包括: 主机、键盘、显示器、驱动器、打印机、绘图机、数字化仪和鼠型定标器, 如图 1-4 所示(IBM 微机绘图系统)。





图 1-4

图 1-5 是 APPLE-II 微机绘图系统, 由五个部分组成: 主机、显示器、软盘驱动器、打印机和绘图机。



图形输入板

绘图仪

主机和显示器

驱动器

打印机

图 1-5

主机: 进行数学运算和数据处理, 控制各种外部设备。

显示器: 显示文字和图形。

打印机: 打印程序清单、表格或图形。

绘图机: 画图。

软盘驱动器: 存储各种信息。

§ 1-3 绘图机的种类

常见的绘图机有平板式和滚筒式两种。

平板式: 纸不动, 笔作 X-Y 方向运动。如 SR-6602, DXY-800、880, SPL-400, WX-4675 等均属平板式绘图仪。

滚筒式: 纸作 X 方向滚动, 笔固定在笔架上, 沿轨道作 Y 方向运动, 二者合成, 实现绘图。如 DMP 系列绘图仪或 SHARP PC-1500、CASIO PB-700 等计算机上所带的袖珍型绘图打印机, 均属滚筒式绘图仪。

§ 1-4 APPLE-II 基本绘图语句

APPLE-II 的绘图, 分为低分辨率与高分辨率两种状态。

一、低分辨率

1. GR

(1) 命令——打开低分辨率画面。此时机器处于低分辨率图形和文本混合工作状态, 坐标原点在屏幕的左上角。每一个低分辨率的页, 能够在屏幕上显示 40 行×48 列的图。行与列的每一个交点, 在屏幕上就是一个小的长方形, 每一页具有 1920(40×48)个交点。

(2) 格式——GR

2. COLOR

(1) 命令——选定颜色。

(2) 格式——COLOR = n

(3) 参数 n: (见表 1-1。)

表 1-1 低分辨率颜色编码

0 黑	4 深绿	8 褐色	12 绿
1 深红	5 深灰	9 桔红	13 黄
2 深蓝	6 蓝	10 灰	14 蓝绿
3 紫	7 淡蓝	11 粉红	15 白

3. PLOT

(1) 命令——在低分辨率画面上, 用目前正在使用的颜色画出一。

(2) 格式——PLOT X, Y

(3) 参数:

X 坐标范围 0~39, Y 坐标范围 0~47(全画面)或 0~39(保留文字幕状态)。

(4) 举例:

10 GR

15 COLOR = 13

20 PLOT 0,0

30 PLOT 0,39

40 PLOT 39,39

50 PLOT 39,0

RUN

显示四个点。

说明:

①第一句话:“10 GR”进入低分辨率状态作图。凡要作图,均要首先设作图状态(低分辨率或高分辨率)。

②第二句话:“15 COLOR = 13”确定颜色,13 表示黄色。

③第三句至第六句:“20 PLOT 0,0”至“50 PLOT 39,0”是显示四个点的指令。

④BASIC 程序的基本结构

一个 BASIC 程序是由若干行组成的。一般规定一行写一个语句,每个语句让计算机执行某一方面的功能。

一个语句可分为三部分:标号、定义符和语句体。

譬如,第一句“10 GR”最前面的 10 是语句标号。每个语句的首部都以数字开始,这个数字称为语句标号(简称标号)。此号码可为 1 至 5 位正整数。计算机按标号由小而大顺序执行各语句。标号一般不连续(如上例中由 10→15,间隔 5;20→30,间隔 10)以便在修改程序时可以增补一些新的语句。标号后面的是语句定义符,它规定计算机执行某项特定的工作。在此,“GR”即绘图指令。不同的机器,绘图指令可能不同。

再譬如,第三句“20 PLOT 0,0”,前面的 20 也是标号(如果不想写 20,也可写成 17 或 19 等),标号后面的 PLOT 是语句定义符,表示画点。在定义符后面的是执行的内容,称为语句体。在此,即要求在(0,0)位置,显示一个点。

4.HLIN

(1)命令——画水平线。

(2)格式——HLIN X1,X2 AT Y

(3)参数:

起点(X1,Y),终点(X2,Y)

(4)举例:

```
10 GR:COLOR=15
```

```
20 HLIN 0,39 AT 0
```

```
RUN
```

在屏幕的顶部画出一条自左到右的水平线。

5.VLIN

(1)命令——画垂直线。

(2)格式——VLIN Y1,Y2 AT X

(3)参数:

起点(X,Y1),终点(X,Y2)。

(4)举例:

```
10 GR:COLOR=15
```

```
20 VLIN 0, 30 AT 15
```

```
RUN
```

以(15,0)为起点,(15,30)为终点,画出一条铅垂线。

上面两例中,冒号“:”表示两句话合并为一句,省一个语句标号。

```

10 GR:COLOR=13
20 HLIN 0,20 AT 0
30 HLIN 0,20 AT 15
40 HLIN 0,20 AT 30
50 VLIN 0,30 AT 10
RUN

```

屏幕上出现一个黄色的“王”字。

小结:

* 低分辨率画面共有两页。当使用 GR 命令后,进入图、文混合页,除了可以绘图外,还可以显示四行文字。如果取消这四行文字,使全屏幕显示图形,可在 GR 之后键入 POKE-16302,0 此时四行文字资料便消失了,进入全屏幕图形页。若要返回图、文混合页,可键入 POKE-16301,0。

* 低分辨率画面由 40×40 或 40×48 个象素组成,光点呈方形,常用于广告画或游戏画面。由于图象粗糙,不宜显示工程图。

二、高分辨率

高分辨率画面也有两页可供使用,一页是图、文混合页,另一页是全屏幕图形页。屏幕显示 192 行 $\times$ 280 列。坐标原点仍在屏幕的左上角,由原点向右为水平轴正方向,由原点向下为竖直轴的正方向。

1. 高分辨率绘图指令

(1) HGR

①命令——高分辨率画面,第 1 页。

②格式——HGR

③注:高分辨率画面第 1 页为图、文混合页,160 行 $\times$ 280 列。屏幕最下方可显示四行文字,如图 1-6。若要全画面,键入 POKE-16302,0。




图 1-6

(2) HGR2

①命令——高分辨率画面,第 2 页。

②格式——HGR2

③注:执行 HGR2 即全画面,如图 1-7。若要恢复底下四行文字,键入 POKE-16301,0。




图 1-7

(3) HCOLOR

①命令——高分辨率状态下确定颜色。

②格式——HCOLOR = n

③参数 n: (见表 1-2)

表 1-2 高分辨率颜色编码

0 黑	2 紫*	4 黑	6 蓝*
1 绿	3 白	5 桔红*	7 白

* 依 TV 而定

(4) HPLOT

①命令——在高分辨率画面上,用目前正在使用的颜色画出一。

②格式——四种形式

HPLOT X,Y 在(X,Y)位置上显示一个点。

HPLOT X1,Y1 TO X2,Y2 自(X1,Y1)至(X2,Y2)画直线。

HPLOT TO X,Y 由前面行号所画过的最后一点,用最近所使用的颜色,画一条直线至(X,Y)点。

HPLOT X1,Y1 TO X2,Y2 TO X3,Y3 (TO……) 自(X1,Y1)起逐点连线。

③举例:

例 1: 点

```
10 HGR2
20 HCOLOR= 3
30 HPLOT 120,60
```

RUN

在(120,60)处出现一白点。

例 2: 水平线

```
10 HGR2
20 HCOLOR= 3
30 HPLOT 0,60 TO 150,60
```

RUN

由(0,60)至(150,60)画一条水平线(图 1-8)

例 3: 竖直线

```
10 HGR2
20 HCOLOR= 3
30 HPLOT 10,40 TO 10,120
```

RUN

由(10,40)至(10,120)画一条竖直线(图 1-9)。

例 4: 斜线

```
10 HGR2
20 HCOLOR= 3
30 HPLOT 10,40 TO 200,120
```

RUN

由(10,40)至(200,120)画一条斜线(图 1-10)

例 5: 三角形

```
10 HGR2
20 HCOLOR= 3
30 HPLOT 10,30 TO 10,70 TO 120,7
  0 TO 10,30
```

RUN

画一直角三角形,顶点坐标为(10,30)、(10,70)、(120,70), (图 1-11)。



图 1-8



图 1-9



图 1-10




图 1-11

例 6: 平行四边形

```
10 HGR2
20 HCOLOR= 3
30 HPL0T 50,50 TO 0,100 TO 150,1
   00 TO 200,50 TO 50,50
```

RUN

画一平行四边形,四个顶点坐标分别为(50,50)、(0,100)、(150,100)、(200,50), (图 1-12)。



图 1-12

例 7: 长方体

```
10 HGR2
20 HCOLOR= 3
30 HPL0T 50,50 TO 0,100 TO 150,1
   00 TO 200,50 TO 50,50
40 HPL0T 0,100 TO 0,150 TO 150,1
   50 TO 150,100
50 HPL0T 200,50 TO 200,100 TO 15
   0,150
```

RUN

画一长方体斜二测(图 1-13)。




图 1-13

例 8: 圆

在平面上,以(140,96)为中心,以 80 为半径画圆(图 1-14)。

分析:

在微机绘图中,对于规则曲线,只要将参数方程写入程序中,即可绘图。所谓参数方程,是将图形中任意一点的 X 坐标与 Y 坐标均用一个参数 I 来表示,X 表示成它与 I 的关系,Y 也表示成它与 I 的关系。最常用的参数是角度,APPLE 机上使用角度时均取弧度

对于以原点为中心的圆的参数方程为:

$$\begin{cases} X = R \cdot \cos I \\ Y = R \cdot \sin I \end{cases}$$

屏幕绘图,坐标原点在左上角,若不加以改造,显示出的图象必是 1/4 圆弧。为此,需移坐标系,将原点移到题设的圆心处(A,B)。代入平移公式:

$$\begin{cases} X = X' + A \\ Y = Y' + B \end{cases} \quad (\text{其中 } X', Y' \text{ 满足圆的参数方程。})$$

根据上述数学模型,编出画圆程序。

程序:

```
10 HGR2
20 HCOLOR= 3
30 A = 140:B = 96
40 FOR T = 0 TO 360
50 I = T * 3.14 / 180
60 X = 80 * COS (I)
70 Y = 80 * SIN (I)
80 HPL0T X + A,Y + B
90 NEXT T
```




图 1-14

说明:

10 打开高分辨率绘图第 2 页。

20 确定绘图颜色(白色)。

30 定圆心坐标(140,96), 令 $A = 140, B = 96$

40 角度 T 的取值从 0° 到 360° 。

50 度转化为弧度。

60~70 圆的参数方程。

80 由方程确定圆周上每一点的坐标, 经平移公式处理后, 显示在屏幕上。

90 角度 T 自动取下一个值, 直至结束。

在掌握画圆的语句之后, 读者应归纳整理各种圆弧的语句, 熟悉这些内容, 对于编程绘图将是有益的。

小结:

* 高分辨率绘图指令一览表:

高分辨率	{	第 1 页 HGR	{	颜色 HCOLOR = 3 (白)
				点 HPLOT X,Y
				HPLOT X1,Y1 TO X2,Y2
				线 HPLOT X1,Y1 TO X2,Y2 TO..... TO Xn,Yn
第 2 页 HGR2	{	HPLOT TO Xi,Yi (以前一点为起点, 与现在点 (Xi,Yi) 连线。若在此之前没有点, 则此指令无效。)		

* 微机绘图与尺规作图相仿, 掌握了画线语句和画圆语句, 就必然能画出各种图形, 关键在于熟悉 BASIC 语言, 提高编程技巧。

第二章 应用 BASIC 语言编写绘图程序

目前,我国的 BASIC 系统大体上可分为单用户基本 BASIC、扩展 BASIC 和多用户分时 BASIC 三大类。其中属于基本 BASIC 语句的只有十七种。但是,为了发挥不同类型的计算机的作用,除了应用基本 BASIC 和扩展 BASIC 外,不同的计算机系统仍有一些特殊的规定。本章主要介绍 APPLE-II 机上所使用的基本 BASIC 语句以及它们在编制绘图程序方面的应用。

BASIC 语言是应用最广泛的一种程序语言,它是从 FORTRAN 语言中提炼、简化而来,程序结构简单,语法浅显易懂,即使是初学者,经过本章的学习定能编写出绘图程序来。

§ 2-1 基本 BASIC 语句简介

基本 BASIC 共十七条语句,现用表 2-1 表示其功能。

表 2-1 基本 BASIC 命令与功能

命令	功 能
READ	从数据区读数
DATA	在数据区存放数据
PRINT	打印数字或字符串
LET	计算并赋值(一般可略之)
GOTO	无条件转移
IF	条件转移
FOR	设置并执行循环
NEXT	循环出口
END	程序终止
INPUT	由键盘即时输入数据
DEF	定义一个函数(自定义函数)
GOSUB	转子程序
RETURN	由子程序返回主程序 GOSUB 下面的语句
RESTORE	恢复数据区初始位置
REM	注释
DIM	说明数组的大小
STOP	使程序暂停

§ 2-2 循环(FOR / NEXT)语句

在计算机绘图中使用得最多的是循环语句。所以,为绘图而学习 BASIC 语言,应从循环语句入手。例如前面学过的圆语句中,就应用了循环(FOR / NEXT)语句。

循环语句是由 FOR 语句作为循环起点, NEXT 语句作为循环终点,它们必须成对出

现。

FOR 语句的一般格式为:

FOR X=A TO B STEP C

其中,X 为循环控制变量;

A 为循环变量初值;

B 为循环变量终值;

C 为循环变量的增量(步长)。

该语句的含义是,循环变量 X 取值从 A 开始到 B 结束,每次间隔(步长)为 C。

NEXT 语句的一般格式为:

NEXT X。

该语句的含义是,执行到此语句时,循环控制变量取下一个 X 值(即在原来 X 上加上一个步长)。当步长为 1 时,“STEP 1”可省去不写。

例 1:用点的移动形成一条 45° 线。

分析:

对于 45° 线来讲,线上任意一点的横、纵坐标应相等,即(I,I)当 I 的取值从 0 到 159,即得到 160 个点,点的轨迹即线。

程序:

```
10 HGR2
20 HCOLOR= 3
30 FOR I = 0 TO 159
40 HPLLOT I,I
50 NEXT I
```

RUN

得图 2-1。

例 2:以点(140,96)为中心,画 16 个同心的正方形。

分析:

如果坐标系的原点定在正方形的中央(如图 2-2),对于边长为 10 的正方形来讲,四个对角顶点的坐标分别为(5,5)、(-5,5)、(-5,-5)、(5,-5),代入画线语句,即可画出此正方形。若要 16 个同中心的正方形,只要让正方形的四个顶点的坐标值按一定倍数增大(或缩小),即: $(5 * K, 5 * K)$ 、 $(-5 * K, 5 * K)$ 、 $(-5 * K, -5 * K)$ 、 $(5 * K, -5 * K)$,当 K 的取值由 1 至 16 时,即可作出 16 个正方形。必须注意的是,APPLE-II 的

程序:

```
10 HGR2 : HCOLOR= 3
20 A = 140:B = 96
30 FOR K = 1 TO 16
40 HPLLOT A + 5 * K,B - 5 * K TO A - 5
  A - 5 * K,B - 5 * K TO A - 5
    * K,B + 5 * K TO A + 5 * K,
      B + 5 * K TO A + 5 * K,B - 5
        * K
50 NEXT K
```

• 10 •

屏幕坐标系原点在左上角,若改为以(140,96)为中心,则需平移坐标系,显然四个顶点的坐标也要代入平移公式。更要注意的是,新坐标系中 Y 坐标与屏幕坐标系中 Y 坐标的增值方向正好相反。所以,当 $A=140, B=96$ 时,四个顶点坐标分别为: $(A+5 * K, B-5 * K)$ 、 $(A-5 * K, B-5 * K)$ 、 $(A-5 * K, B+5 * K)$ 、 $(A+5 * K, B+5 * K)$ 。




图 2-1




图 2-2




图 2-3

例 3: 画八个同心圆。

分析:

通过前面的学习,我们已经知道,画圆的程序是——

```

10  HGR2: HCOLOR = 3
20  A = 140: B = 96
25  R = 80
30  FOR T = 0 TO 360
40  I = T * 3.14 / 180
50  X = R * COS(I) : Y = R * SIN(I)
60  HPOINT A+X, B-Y
70  NEXT T

```

执行这个程序,得到的是以(140,96)为圆心,以 80 为半径的圆。改变半径的数值,就可以得到又一个圆。显然,若对半径循环赋值,就能画出同心圆。

程序:

```

10 HGR2 : HCOLOR= 3
20 A = 140:B = 96
25 FOR R = 10 TO 80 STEP 10
30 FOR T = 0 TO 360
40 I = T * 3.14 / 180
50 X = R * COS (I):Y = R * SIN
    (I)
60 HPLLOT X + A,B - Y
70 NEXT T
80 NEXT R

```




图 2-4

说明:

(1) 在 FOR / NEXT 循环中, 可加入另外的 FOR / NEXT 循环, 构成多重循环。如该例中:

```

      25 FOR R=10 TO 80 STEP 10
          |
          | 循环②
          |
          |----- 30 FOR T=0 TO 360
                  |
                  | 循环①
                  |
                  |----- 70 NEXT T
                          |
                          |----- 80 NEXT R

```

这就是双重循环。

使用多重循环必须注意分清循环层次,可以并列,不能相交。并列的循环个数不限,但嵌套的循环重数在某些计算机上不能超过八层。APPLE-II 机上允许有 10 重循环。执行多重循环的顺序是,最内层先循环,由内向外,逐层循环。

如：

```
FOR X
FOR Y
:
NEXT Y
NEXT X
```

正 确

```
FOR X
FOR Y
:
NEXT X
NEXT Y
```

错误

(2)STEP 代表步长,即循环变量取值的间隔。当变量由小至大,步长为正值;当变量由大至小,步长为负值。如上例中 25 句若改为:

```
25 FOR R=80 TO 10 STEP -10
```

其中步长就是-10。

一般当步长为 1 时,“STEP 1”可省略;反之,若不写步长,即默认步长为 1。注意,若步长为-1 时,一定要写上“STEP -1”。

例 4:画多边形

分析:

画圆的程序中,循环变量 T 的取值是每隔 1° 取一个值。若改为每隔 60° 取一个值,即得六边形;每隔 72° 取一个值,即得五边形。所以,只要在画圆的程序中,加上步长,就能得到多边形。

下列程序 1,STEP 60,画六边形;程序 2,STEP 72,画五边形。

程序 1:六边形

```
10 HGR2
20 HCOLOR= 3
30 A = 140:B = 96
32 HPLOT 80 + A,B
40 FOR T = 0 TO 360 STEP 60
50 I = T * 3.14 / 180
60 X = 80 * COS (I)
70 Y = 80 * SIN (I)
80 HPLOT TO X + A,Y + B
90 NEXT T
```




图 2-5

程序 2:五边形

```
10 HGR2
20 HCOLOR= 3
30 A = 140:B = 96
32 HPLOT 80 + A,B
40 FOR T = 0 TO 360 STEP 72
50 I = T * 3.14 / 180
60 X = 80 * COS (I)
70 Y = 80 * SIN (I)
80 HPLOT TO X + A,B - Y
90 NEXT T
```




图 2-6

注意:

画多边形时,要用 HPLOT TOXi, Yi 画线语句,使用该语句时,先要用 HPLOT Xa, Ya 显示点,否则无效。即:

```
30 A=140:B=96
32 HPLOT 80+A,B——显示起点位置。
80 HPLOT TO X+A,B-Y——由起点开始,逐点连线。
```

两句配合使用

§ 2-3 输入(INPUT)语句

INPUT 语句是输入语句中的一种,又称问答式输入或键盘输入语句。对需要经常改

变数值的参数,最好用 INPUT 语句输入,这样可使程序通用、灵活。

譬如,我们可以用输入 (INPUT) 语句来改写画圆的程序,使之对于任意圆心位置、任意半径的圆,该程序都通用。

例 1: 画圆的通用程序

程序:

```
10 INPUT A,B,R
20 HGR2 : HCOLOR= 3
30 FOR T = 0 TO 360
40 I = T * 3.14 / 180
50 X = R * COS (I):Y = R * SIN
  (I)
60 HPLOT A + X,B - Y
70 NEXT T
```




图 2-7

其中 A、B 为圆心的横、纵坐标, R 为圆的半径。

输入不同的数值,即得不同位置,不同大小的圆(如图 2-7)。

INPUT 语句是人机对话方式,在程序运行时,计算机遇到 INPUT 时,就停下来,在屏上显示一个问号“?”,这就是机器询问用户的意思,此时可由键盘输入一个数值到程序里(该数值被送到存储器中该变量的单元里),然后继续执行程序。

如上例 1,当按 RUN 后,屏上显示“?”,即询问圆心的横坐标 A=?,这时可按用户的要求输入一个数值(如输入 50),然后回车。这时屏上显示“??”,即询问圆心的纵坐标 B=?,按用户的要求输入另一个数值(如输入 60),然后回车。这时屏上显示“??”,即询问圆的半径 R=?,按用户的要求再输入一个数值(如输入 30),然后回车。这时在屏幕的左上角出现一个小圆。若要绘制另一个圆,只要用键盘输入另外三个数,则又画一个圆。程序可以重复使用。

全部执行过程是:

按 **RUN** **RETURN** 显示?

键入 50 按 **RETURN** 显示??

键入 60 按 **RETURN** 显示??

键入 30 按 **RETURN** 开始作图

如果采取 INPUT“A,B,R=”;A,B,R 形式,执行时:

按 **RUN** **RETURN** 显示 A,B,R=

继等号之后,键入 50,60,30 按 **RETURN** 则出图。

注意:变量之间必须用逗号分隔,最后一个变量的后面,不能有标点符号。

注:如果用户打入的数据少于 INPUT 语句中一系列变量的个数,那么在屏幕上将不断出现两个问号“??”,提示用户逐个赋值,直到每个变量都得到数值为止。

例 1 中的第一句也可改写成如下形式:

```
10 INPUT“A,B,R=”;A,B,R
```

这种形式与 INPUT A,B,R 所表达的意思一样,只是在执行中屏上显示的内容不同。